QT       += core gui

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets network charts multimedia

CONFIG += c++11


# The following define makes your compiler emit warnings if you use
# any Qt feature that has been marked deprecated (the exact warnings
# depend on your compiler). Please consult the documentation of the
# deprecated API in order to know how to port your code away from it.
DEFINES += QT_DEPRECATED_WARNINGS
win32-msvc*:QMAKE_CXXFLAGS += /utf-8

# You can also make your code fail to compile if it uses deprecated APIs.
# In order to do so, uncomment the following line.
# You can also select to disable deprecated APIs only up to a certain version of Qt.
#DEFINES += QT_DISABLE_DEPRECATED_BEFORE=0x060000    # disables all the APIs deprecated before Qt 6.0.0

INCLUDEPATH += $$PWD/component
INCLUDEPATH += $$PWD/page
INCLUDEPATH += $$PWD/utils

CONFIG      += warn_off

## boost
INCLUDEPATH += $$PWD/third_lib/boost_1_76
DEPENDPATH += $$PWD/third_lib/boost_1_76

## ui lib
INCLUDEPATH += $$PWD/third_lib/design
DEPENDPATH += $$PWD/third_lib/design

#表示64位的构建套件
contains(QT_ARCH, x86_64) {
MOC_DIR     = $$PWD/temp_64/moc
RCC_DIR     = $$PWD/temp_64/rcc
UI_DIR      = $$PWD/temp_64/ui
OBJECTS_DIR = $$PWD/temp_64/obj
#指定编译生成的可执行文件到bin目录
DESTDIR     = $$PWD/temp_64/bin
CONFIG(release, debug|release){
LIBS += -L$$PWD/third_lib/design/ -lhelp_VS2015_64
#boost
LIBS += -L$$PWD/third_lib/boost_1_76/lib/ -llibboost_date_time-vc142-mt-x64-1_76
}else
{
#ui
win32:LIBS += -L$$PWD/third_lib/design/ -lhelp_VS2015_64d
#boost
LIBS += -L$$PWD/third_lib/boost_1_76/lib/ -llibboost_date_time-vc142-mt-gd-x64-1_76
}
} else {
MOC_DIR     = $$PWD/temp_32/moc
RCC_DIR     = $$PWD/temp_32/rcc
UI_DIR      = $$PWD/temp_32/ui
OBJECTS_DIR = $$PWD/temp_32/obj
#指定编译生成的可执行文件到bin目录
DESTDIR     = $$PWD/temp_32/bin
CONFIG(release, debug|release){
LIBS += -L$$PWD/third_lib/design/ -lhelp_VS2019_32
#boost
LIBS += -L$$PWD/third_lib/boost_1_76/lib/ -llibboost_date_time-vc142-mt-x32-1_76
}else
{
#ui
win32:LIBS += -L$$PWD/third_lib/design/ -lhelp_VS2019_32d
#boost
LIBS += -L$$PWD/third_lib/boost_1_76/lib/ -llibboost_date_time-vc142-mt-gd-x32-1_76
}
}

# websocketpp
INCLUDEPATH += $$PWD/websocketpp


# 松动检查
win32: QMAKE_CXXFLAGS_RELEASE -= -Zc:strictStrings
win32: QMAKE_CFLAGS_RELEASE -= -Zc:strictStrings
win32: QMAKE_CFLAGS -= -Zc:strictStrings
win32: QMAKE_CXXFLAGS -= -Zc:strictStrings

SOURCES += \
    component/DrawForm.cpp \
    component/about.cpp \
    component/addaifaceform.cpp \
    component/addchannelform.cpp \
    component/algo_arm_form.cpp \
    component/algo_cam_h_form.cpp \
    component/algo_cam_l_form.cpp \
    component/algo_dzft_form.cpp \
    component/iplinkdialog.cpp \
    component/netparamform.cpp \
    component/netparamplusform.cpp \
    component/reviseaifaceform.cpp \
    component/revisechannelform.cpp \
    main.cpp \
    mainwindow.cpp \
    page/CommonSetForm.cpp \
    page/DevUpForm.cpp \
    page/SnapForm.cpp \
    page/aifaceform.cpp \
    page/algoform.cpp \
    page/algosetform.cpp \
    page/cameraform.cpp \
    page/coalform.cpp \
    page/complexsetform.cpp \
    utils/AiWebsocket.cpp \
    utils/VoiceTalk.cpp \
    utils/httpclient.cpp \
    utils/iconhelper.cpp \
    utils/maskwidget.cpp \
    utils/myapp.cpp \
    utils/mylabel.cpp \
    utils/singleton.cpp

HEADERS += \
    AiPtuStruct.h \
    component/DrawForm.h \
    component/about.h \
    component/addaifaceform.h \
    component/addchannelform.h \
    component/algo_arm_form.h \
    component/algo_cam_h_form.h \
    component/algo_cam_l_form.h \
    component/algo_dzft_form.h \
    component/iplinkdialog.h \
    component/netparamform.h \
    component/netparamplusform.h \
    component/reviseaifaceform.h \
    component/revisechannelform.h \
    mainwindow.h \
    page/CommonSetForm.h \
    page/DevUpForm.h \
    page/SnapForm.h \
    page/aifaceform.h \
    page/algoform.h \
    page/algosetform.h \
    page/cameraform.h \
    page/coalform.h \
    page/complexsetform.h \
    third_lib/design/switchbutton.h \
    utils/AiWebsocket.h \
    utils/VoiceTalk.h \
    utils/checkboxheaderview.h \
    utils/httpclient.h \
    utils/iconhelper.h \
    utils/maskwidget.h \
    utils/myapp.h \
    utils/myhelper.h \
    utils/mylabel.h \
    utils/singleton.h

FORMS += \
    component/DrawForm.ui \
    component/about.ui \
    component/addaifaceform.ui \
    component/addchannelform.ui \
    component/algo_arm_form.ui \
    component/algo_cam_h_form.ui \
    component/algo_cam_l_form.ui \
    component/algo_dzft_form.ui \
    component/iplinkdialog.ui \
    component/netparamform.ui \
    component/netparamplusform.ui \
    component/reviseaifaceform.ui \
    component/revisechannelform.ui \
    mainwindow.ui \
    page/CommonSetForm.ui \
    page/DevUpForm.ui \
    page/SnapForm.ui \
    page/aifaceform.ui \
    page/algoform.ui \
    page/algosetform.ui \
    page/cameraform.ui \
    page/coalform.ui \
    page/complexsetform.ui

RESOURCES += \
    res.qrc

RC_ICONS = ptu.ico.
VERSION = 1.0.3


INCLUDEPATH += $$PWD xlsx
include(xlsx/qtxlsx.pri)
